Polynomials and Pretzels: The Art of Graphing

Steven J. Wilson

Types of roots of polynomials:



Single

$(x - a)$



Even

$(x - a)^n$, n even



Odd, $n > 1$

$(x - a)^n$, $n > 1$ & odd

1. How can the complex roots of a polynomial be found from a graph?
2. What is the effect on the graph when the variable y in an equation is replaced by $y - c$?, or by cy ?
3. How are the roots and the asymptotes of the graph of a rational function and its equation related? (A rational function is a function of the form $y = P(x)/Q(x)$, where $P(x)$ and $Q(x)$ are polynomials.)

Some types of singularities:

Node

Cusp

Double Cusp (Tacnode, Pt. of Osculation)

Vertical Tangent

Isolated Point (Acnode)

A <u>semifunction</u> is a relation (not a function) of the form $y^2 = f(x)$, where $f(x)$ is a function of x . Since a semifunction can be easily solved for y , a graphing calculator or computer software package can be used to graph them.

4. Under what conditions will a semifunction possess a particular type of singularity?
5. Are there other types of singularities a semifunction can have?
6. How many different types of graphs can be obtained from a semifunction of an n-th degree polynomial?
7. How is a graph affected when the y variable in its equation is replaced by y^2 , y^3 , $|y|$, $\frac{1}{y}$, $\sin y$?

Both of the computer programs listed below graph any relation of the variables x and y . To use the program:

1. Move all terms of the equation to the right side of the equal sign.
2. Replace the zero on the left hand side of the equation by the variable z .
3. Enter the equation into line 270 (GRAFANY1) or line 680 (GRAFANY2).
4. Adjust the scale of the display in line 170 (GRAFANY1) or line 190 (GRAFANY2).
5. Run it.

These are slow programs, since high resolution was desired at any cost. If an equation can be solved for y , a canned software package should be used instead. GRAFANY2 will generally give satisfactory results for uncomplicated graphs of complicated equations, but portions of the graph can be lost if the graph is itself quite complicated.

```
100 'GRAFANY1 - a high-resolution slow-speed 2-variable relation grapher.
110 'Execution times average 2-7 hours.
120 'Programmed for a Tandy 1000, Jan. 14, 1991, by Steven J. Wilson.
130 '
140 'initialization
150 '
160 DIM F(199)                                  'function memory
170 X2=3: X1=-X2: Y1=.75*X1: Y2=.75*X2          'max & min coordinates
180 SCREEN 2                                    'high resolution
190 CLS: KEY OFF: BEEP ON
200 '
210 'check every pixel for presence of graph
220 '
230 FOR I=0 TO 639
240  FOR J=0 TO 199
250   X=X1+I*(X2-X1)/639                        'screen to Cartesian coordinates
260   Y=Y2-J*(Y2-Y1)/199
270   Z=(Y^2-2)^2-(X+2)*(X-1)^2                 'equation of a pretzel
280   IF J=0 THEN 300                           'error trap
290   IF Z*F(J-1)<=0 AND ABS(Z-F(J-1))<100 THEN PSET (I,J) 'vertical sign change
300   IF I=0 THEN 320                           'error trap
310   IF Z*F(J)<=0 AND ABS(Z-F(J))<100 THEN PSET (I,J)   'horizontal sign change
320   F(J)=Z                                    'update memory
330  NEXT J
340 NEXT I
350 '
360 'end of computation
370 '
380 BEEP
390 DEF SEG = &HB800                            'location of video memory
400 BSAVE "B:PIX",0,16384                       'save screen display on disk
410 '   to recall screen display from disk, type:
420 '   SCREEN 2: DEF SEG = &HB800: BLOAD "B:PIX"
430 END
```

```
100 'GRAFANY2 - a mostly-high-resolution medium-speed 2-variable
110 'relation grapher.  Execution times average 20-60 minutes.
120 'The fine detail of the graph of a complicated equation may be lost.
130 'Programmed for a Tandy 1000, Jan. 28, 1991, by Steven J. Wilson.
140 '
150 'initialization
160 '
170 SCREEN 2                                'high resolution
180 CLS: KEY OFF: BEEP ON
190 X2=3: X1=-X2: Y1=.75*X1: Y2=.75*X2  'max & min coordinates
200 G1=5: G2=4                              'grid size for speed
210 DIM F(G2)                               'memory inside grid
220 DIM G(INT(200/G2)+1)                    'memory for grid
230 '
240 'check grid for possible occupied squares
250 '
260 FOR GI = 0 TO 639 STEP G1
270 FOR GJ = 0 TO 199 STEP G2
280   X=X1+GI*(X2-X1)/639                   'screen to Cartesian coordinates
290   Y=Y2-GJ*(Y2-Y1)/199
300   GOSUB 680                             'evaluate equation
310   IF GI=0 OR GJ=0 THEN 350              'initial row and column
320   'line 330 identifies possible occupied squares
330   IF Z*G(GJ/G2)<=0 OR Z*G(GJ/G2-1)<=0 OR Z*GT<=0 THEN GOSUB 500
340   G(GJ/G2-1)=GT                         'update memory
350   GT=Z
360 NEXT GJ
370 NEXT GI
380 '
390 'end of computation
400 '
410 BEEP
420 DEF SEG = &HB800                        'location of video memory
430 BSAVE "B:PIX",0,16384                   'save screen display on disk
440 '    to recall screen display from disk, type:
450 '    SCREEN 2: DEF SEG = &HB800: BLOAD "B:PIX"
460 END
470 '
480 'check possible occupied squares for points satisfying equation
490 '
500 FOR I = GI-G1 TO GI
510 FOR J = GJ-G2 TO GJ
520   X=X1+I*(X2-X1)/639                    'screen to Cartesian coordinates
530   Y=Y2-J*(Y2-Y1)/199
540   GOSUB 680                             'evaluate equation
550   T=J-GJ+G2                             'location in small grid
560   IF T=0 OR I-GI+G1=0 THEN 610          'initial row and column of square
570   IF J<=0 OR J>=200 THEN 590            'off screen
580   IF Z*F(T-1)<=0 AND ABS(Z-F(T-1))<100 THEN PSET (I,J) 'vertical sign change
590   IF I<=0 OR I>=640 THEN 610            'off screen
600   IF Z*F(T)<=0 AND ABS(Z-F(T))<100 THEN PSET (I,J)   'horizontal sign change
610   F(T)=Z                                'update memory
620 NEXT J
630 NEXT I
640 RETURN
650 '
660 'the equation to be graphed, written in the form  Z = F(x,y)
670 '
680 Z=(Y^2-2)^2-(X+2)*(X-1)^2               'equation of a pretzel
690 RETURN
```


POLYNOMIALS
AND PRETZELS:
THE ART OF
GRAPHING

The Art of Graphing

1. Given an equation, how can we find a graph?

2. Given a graph, how can we find an equation?



$y = x$



$y = x^2$

$y = x^3$

POWER FUNCTIONS

$y = x^n$



$y = x^4$



$y = x^5$

POLYNOMIAL FUNCTIONS



$y = x\,(x - 1)$



$y = x\,(x + 1)\,(x - 1)$



$y = x\,(x + 1)\,(x - 1)\,(x - 2)$

What is the equation of this polynomial function?

$$y = -x^3 (x + 2) (x + 1.5)^2 (x - 1)^2$$

$$y = -(x + 2)(x + 1)(x - 1)(x - 2)$$





$$y^2 = -(x + 2)(x + 1)(x - 1)(x - 2)$$



$$x^2 + (y - 1)^2 = 1$$



$$x^2 + \left(y^2 - 1 \right)^2 = 1$$

$y = x^3 - 3x$



$y = x^3 - 3x + 2$



$y^2 = x^3 - 3x + 2$



$(y - 2)^2 = x^3 - 3x + 2$



$(y^2 - 2)^2 = x^3 - 3x + 2$



HOW TO GRAPH

A PRETZEL

SEMIPRETZELS



$$\left\{\left[\left(y^2 - 2\right)^2 - 2\right]\right\}^2 = x^3 - 3x + 2$$



$$\left(y^2 - 2\right)^2 = x^2\left(x^2 - 3\right)^2$$

OCTOPRETZEL



$$\left|\left|\left|\left|y^2 - \sqrt{2 + \sqrt{2 + 2\sqrt{5}}}\right|^2 - \sqrt{2 + 2\sqrt{5}}\right|^2 - 2\right|^2$$

$$= \left(-x^2 + 3\right)^3 - 3\left(-x^2 + 3\right) + 2$$

TRIGONOMETRIC PRETZELS



$$\left[\left(2 + 2\sqrt{5}\right)\sin^2 y - 2\right]^2 = \left(3\sin x\right)^3 - 3\left(3\sin x\right) + 2$$